





## Adafruit 4-Channel ADC Breakouts

Created by Bill Earl



Last updated on 2014-11-30 10:00:14 AM EST


# Guide Contents

Guide Contents	2
Overview	4
ADS1115 Features:	4
ADS1015 Features:	5
Assembly and Wiring	6
Assembly:	6
Prepare the header strip	6
Position the breakout board	6
Solder!	6
Wiring:	7
Power	7
I2C Connections	7
I2C "Classic"	8
I2C Addressing	8
Multiple Boards	9
Signal Connections	11
Single Ended vs. Differential Inputs:	11
Which should I use?	11
Single Ended Connections:	11
Differential Connections:	12
Programming	13
Construction and Initialization:	13
Single Ended Conversion:	13
Differential Conversion:	14
Comparator Operation:	15
Adjusting Gain	16
Example	17
Downloads	18
Library:	18
Board Files:	18



# Overview

---



The ADS1115 and ADS1015 4-channel breakout boards are perfect for adding high-resolution analog to digital conversion to any microprocessor-based project. These boards can run with power and logic signals between 2v to 5v, so they are compatible with all common 3.3v and 5v processors. As many of 4 of these boards can be controlled from the same 2-wire I2C bus, giving you up to 16 single-ended or 8 differential channels. A programmable gain amplifier provides up to x16 gain for small signals.

These two boards are very similar, differing only in resolution and speed. The ADS1115 has higher resolution and the ADS1015 has a higher sample rate.

## ADS1115 Features:

---

- Resolution: 16 Bits
- Programmable Sample Rate: 8 to 860 Samples/Second
- Power Supply/Logic Levels: 2.0V to 5.5V
- Low Current Consumption: Continuous Mode: Only 150 $\mu$ A Single-Shot Mode: Auto Shut-Down
- Internal Low-Drift Voltage Reference
- Internal Oscillator
- Internal PGA: up to x16
- I2C Interface: 4-Pin-Selectable Addresses

- Four Single-Ended or 2 Differential Inputs
- Programmable Comparator

## ADS1015 Features:

---

- Resolution: 12 Bits
- Programmable Sample Rate: 128 to 3300 Samples/Second
- Power Supply/Logic Levels: 2.0V to 5.5V
- Low Current Consumption: Continuous Mode: Only 150 $\mu$ A Single-Shot Mode: Auto Shut-Down
- Internal Low-Drift Voltage Reference
- Internal Oscillator
- Internal PGA: up to x16
- I2C Interface: 4-Pin-Selectable Addresses
- Four Single-Ended or 2 Differential Inputs
- Programmable Comparator


# Assembly and Wiring

## Assembly:


The board comes with all surface-mount parts pre-soldered. For breadboard use, the included header-strip should be soldered on:



**Prepare the header strip**  
Cut the supplied header strip to length and insert it long-pins-down in your breadboard to hold it for soldering.



**Position the breakout board**  
Place the breakout board on the header pins.




**Solder!**  
Solder each pin for a good electrical connection.

## Wiring:

### Power


First connect VDD and GND. These boards will work with either a 3.3v or a 5v supply. The diagram below shows connection to the Arduino 5v pin.

The absolute maximum analog input voltage is  $VDD + 0.3v$ . To avoid damage to the chip, do not attempt to measure voltages greater than VDD.




### I2C Connections

I2C requires just 2 pins to communicate. These can be shared with other I2C devices. For R3 and later Arduinos (including MEGA and DUE models), connect SDA->SDA and SCL->SCL.



## I2C "Classic"

For older Arduino boards without dedicated SDA and SCL pins, connect as shown below. (For older Arduino Megas, SDA and SCL are on pins 20 and 21)




## I2C Addressing

The ADS11x5 chips have a base 7-bit I2C address of 0x48 (1001000) and a clever addressing scheme that allows four different addresses using just one address pin (named **ADR** for ADdRes). To program the address, connect the address pin as follows:




- 0x48 (1001000) ADR -> GND
- 0x49 (1001001) ADR -> VDD
- 0x4A (1001010) ADR -> SDA
- 0x4B (1001011) ADR -> SCL

The following diagram shows one board addressed as 0x48:




## Multiple Boards

By assigning each board a different address, up to 4 boards can be connected as below:



# Signal Connections



## Single Ended vs. Differential Inputs:

The ADS1x15 breakouts support up to 4 Single Ended or 2 Differential inputs.

**Single Ended** inputs measure the voltage between the analog input channel (A0-A3) and analog ground (GND).

**Differential** inputs measure the voltage between two analog input channels. (A0&A1 or A2&A3).

## Which should I use?

Single ended inputs give you twice as many inputs. So why would you want to use differential inputs?


Single ended inputs can, by definition, only measure positive voltages. Without the sign bit, you only get an effective 15 bit resolution.

In addition to providing the full 16 bits of resolution and the ability to measure negative voltages, Differential measurements offer more immunity from electromagnetic noise. This is useful when using long signal wires or operating in an electrically noisy environment. This is also desirable when dealing with small signals requiring high gain, since the gain will amplify the noise as well as the signal.

## Single Ended Connections:


Connect the signal wire to one of the analog input channels (A0 - A3). Connect the ground

wire to GND. This diagram shows how to connect an ADXL335 to for measurement of the X, Y and Z axis on analog channels A0, A1 and A2.



## Differential Connections:

Differential measurements use a pair of input pins, either A0&A1 or A2&A3. The following diagram shows connections for differential measurement of the battery voltage on a LiPo charger board.



All input signals to these devices must be between ground potential and VCC. If your source signal produces negative voltages, they must be offset to fall within the GND to VCC range of the ASD1x15.

# Programming

The Adafruit\_ADS1x15 library supports both single-ended and differential readings as well as comparator operations on both the ADS1015 and ADS1115 breakout boards. The library uses the wiring library for I2C communication, so wiring.h must be included.

## Construction and Initialization:

### **Adafruit\_ADS1015();**

Construct an instance of an ADS1015 with the default address (0x48)

### **Adafruit\_ADS1015(uint8\_t addr);**

Construct an instance of an ADS1015 with the specified address (0x48 - 0x4B)

### **Adafruit\_ADS1115();**

Construct an instance of an ADS1115 with the default address (0x48)

### **Adafruit\_ADS1115(uint8\_t addr);**

Construct an instance of an ADS1115 with the specified address (0x48 - 0x4B)

### **void begin(void);**

Initialize the ADC for operation.

### **Example:**

The following examples assume an ADS1015 and use a 3 mV/bit scaling factor. For the higher-resolution ADS1115, the scaling factor would be 188uV/bit.

```
#include <Wire.h>
#include <Adafruit_ADS1015.h>

Adafruit_ADS1015 ads1015; // Construct an ads1015 at the default address: 0x48
Adafruit_ADS1115 ads1115(0x49); // construct an ads1115 at address 0x49

void setup(void)
{
  ads1015.begin(); // Initialize ads1015
  ads1115.begin(); // Initialize ads1115
}
```

## Single Ended Conversion:

### **uint16\_t readADC\_SingleEnded(uint8\_t channel);**

Perform a single-ended analog to digital conversion on the specified channel.

## Example:

```
#include <Wire.h>
#include <Adafruit_ADS1015.h>

Adafruit_ADS1015 ads1015;

void setup(void)
{
  Serial.begin(9600);
  Serial.println("Hello!");

  Serial.println("Getting single-ended readings from AIN0..3");
  Serial.println("ADC Range: +/- 6.144V (1 bit = 3mV)");
  ads1015.begin();
}

void loop(void)
{
  int16_t adc0, adc1, adc2, adc3;

  adc0 = ads1015.readADC_SingleEnded(0);
  adc1 = ads1015.readADC_SingleEnded(1);
  adc2 = ads1015.readADC_SingleEnded(2);
  adc3 = ads1015.readADC_SingleEnded(3);
  Serial.print("AIN0: "); Serial.println(adc0);
  Serial.print("AIN1: "); Serial.println(adc1);
  Serial.print("AIN2: "); Serial.println(adc2);
  Serial.print("AIN3: "); Serial.println(adc3);
  Serial.println(" ");

  delay(1000);
}
```

## Differential Conversion:

### **int16\_t readADC\_Differential\_0\_1(void);**

Perform a differential analog to digital conversion on the voltage between channels 0 and 1.

### **int16\_t readADC\_Differential\_2\_3(void);**

Perform a differential analog to digital conversion on the voltage between channels 2 and 3.

## Example:

```

#include <Wire.h>
#include <Adafruit_ADS1015.h>

Adafruit_ADS1015 ads1015;

void setup(void)
{
  Serial.begin(9600);
  Serial.println("Hello!");

  Serial.println("Getting differential reading from AIN0 (P) and AIN1 (N)");
  Serial.println("ADC Range: +/- 6.144V (1 bit = 3mV)");
  ads1015.begin();
}

void loop(void)
{
  int16_t results;

  results = ads1015.readADC_Differential_0_1();
  Serial.print("Differential: "); Serial.print(results); Serial.print(" "); Serial.print(results * 3); Serial.println("mV");

  delay(1000);
}

```

## Comparator Operation:

Comparator mode allows you to compare an input voltage with a threshold level and generate an alert signal (on the ALRT pin) if the threshold is exceeded. This pin can be polled with a digital input pin, or it can be configured to generate an interrupt.

**void startComparator\_SingleEnded(uint8\_t channel, int16\_t threshold);**

Set the threshold and channel for comparator operation.

**int16\_t getLastConversionResults();**

Get the last conversion result and clear the comparator.

### Example:

```

#include <Wire.h>
#include <Adafruit_ADS1015.h>

Adafruit_ADS1015 ads1015;

void setup(void)

```

```

{
  Serial.begin(9600);
  Serial.println("Hello!");

  Serial.println("Single-ended readings from AIN0 with >3.0V comparator");
  Serial.println("ADC Range: +/- 6.144V (1 bit = 3mV)");
  Serial.println("Comparator Threshold: 1000 (3.000V)");
  ads1015.begin();

  // Setup 3V comparator on channel 0
  ads1015.startComparator_SingleEnded(0, 1000);
}

void loop(void)
{
  int16_t adc0;

  // Comparator will only de-assert after a read
  adc0 = ads1015.getLastConversionResults();
  Serial.print("AIN0: "); Serial.println(adc0);

  delay(100);
}

```

## Adjusting Gain

To boost small signals, the gain can be adjusted on the ADS1x15 chips in the following steps:

- **GAIN\_TWOTHIRDS** (for an input range of +/- 6.144V)
- **GAIN\_ONE** (for an input range of +/-4.096V)
- **GAIN\_TWO** (for an input range of +/-2.048V)
- **GAIN\_FOUR** (for an input range of +/-1.024V)
- **GAIN\_EIGHT** (for an input range of +/-0.512V)
- **GAIN\_SIXTEEN** (for an input range of +/-0.256V)

### **adsGain\_t getGain(void)**

Reads the current gain value (default = 2/3x)

```
adsGain_t gain = getGain();
```

### **void setGain(adsGain\_t gain)**

Sets the gain for the ADS1x15



```
ads1015.setGain(GAIN_TWOTHIRDS); // 2/3x gain +/- 6.144V 1 bit = 3mV (default)

// ads1015.setGain(GAIN_ONE); // 1x gain +/- 4.096V 1 bit = 2mV
// ads1015.setGain(GAIN_TWO); // 2x gain +/- 2.048V 1 bit = 1mV
// ads1015.setGain(GAIN_FOUR); // 4x gain +/- 1.024V 1 bit = 0.5mV
// ads1015.setGain(GAIN_EIGHT); // 8x gain +/- 0.512V 1 bit = 0.25mV
// ads1015.setGain(GAIN_SIXTEEN); // 16x gain +/- 0.256V 1 bit = 0.125mV
```

## Example

If we had an analog sensor with an output voltage  $\sim 1\text{V}$  (a TMP36, for example), we could set the gain on the ADC to **GAIN\_FOUR**, which would give us a  $\pm 1.024\text{V}$  range. This would push the  $1\text{V}$  input signal over the entire 12-bit or 16-bit range of the ADC, compared to the very limited range  $1\text{V}$  would cover without adjusting the gain settings

```
// Set the gain to 4x, for an input range of +/- 1.024V
// 1-bit = 0.5V on the ADS1015 with this gain setting
ads1015.setGain(GAIN_FOUR);
```

# Downloads

---

## Library:

[ADS1x15 Library for Arduino \(http://adafru.it/aSt\)](http://adafru.it/aSt)

## Board Files:

[Board Files and Schematics \(http://adafru.it/aSu\)](http://adafru.it/aSu)

## Data Sheets:

[ADS1015 Data Sheet \(http://adafru.it/aSv\)](http://adafru.it/aSv)

[ADS1115 Data Sheet \(http://adafru.it/aSw\)](http://adafru.it/aSw)